

# Metodo delle Secanti

Enrico Bertolazzi

## [-] Introduzione

**Scopo:** data una approssimazione  $x^*$  di uno zero di  $f(x)$  migliora la approssimazione applicando il metodo delle secanti

## [-] Carica le librerie

```
> restart;  
with(plots):  
Warning, the name changecoords has been redefined
```

## [-] Procedura Secanti

```
> secanti_step := proc (xx0, xx1, f, epsi, max_iter)  
  local i, x0, x1, x2, f0, f1, S, dx ;  
  x0 := xx0 ;  
  x1 := xx1 ;  
  f0 := f(x0) ;  
  f1 := f(x1) ;  
  for i from 1 to max_iter do  
    S := evalf((f1-f0)/(x1-x0),12);  
    dx := evalf(f1/S,8) ;  
    x2 := x1 - dx ;  
    if abs(dx) < epsi then break ; end if ;  
    x0 := x1 ;  
    x1 := x2 ;  
    f0 := f1 ;  
    f1 := f(x2) ;  
  end do;  
  return x2 ;  
end:
```

## [-] Procedura Secanti con grafica

```
> secanti_graph := proc (xx0, xx1, f, epsi, max_iter, xmin, xmax)  
  local i, x0, x1, x2, f0, f1, S, dx, vert, seca, A, B, C ;  
  x0 := xx0 ;  
  x1 := xx1 ;  
  f0 := f(x0) ;
```

```

f1      := f(x1) ;
seca    := [ ] ;
vert    := [ [ [x0,0], [x0,f0] ] ] ;

for i from 1 to max_iter do

    vert := [ op(vert), [ [x1,f1], [x1,0] ] ] ;

    S     := evalf((f1-f0)/(x1-x0),8);
    dx    := evalf(f1/S,8) ;
    x2    := x1 - dx ;

    seca := [ op(seca), [ [x0,f0], [x1,f1], [x2,0] ] ] ;

    if abs(dx) < epsi then break ; end if ;
    x0    := x1 ;
    x1    := x2 ;
    f0    := f1 ;
    f1    := f(x2) ;

end do ;
A := plot(seca, color=red):
B := plot(vert, color=blue):
C := plot(f(x), 'x' = xmin..xmax, color=green, thickness=2):
display({A,B,C});
end:

```

## - Esempio d'uso

> Definisce alcuni parametri comuni

```

> epsi      := 1e-5 :
max_iter    := 5 :

```

> Esempio di loop complicato

```

> f := (x) -> (x/(x^2)^(1/2))*(x^2)^(1/4) ;

```

Risolve il problema e stampa i risultati

```

> secanti_step(3,2,f,epsi,max_iter) ;

```

Risolve il problema e grafica la procedura


```

> secanti_graph(3,2,f,epsi,max_iter,-5,5) ;

```

$$f := x \rightarrow \frac{x(x^2)^{(1/4)}}{\sqrt{x^2}}$$

-0.21602634



> Esempio di convergenza monotona

> **f := (x) -> x^2-3 ;**

Risolve il problema e stampa i risultati


> **secanti\_step(3,2,f,epsi,max\_iter) ;**

Risolve il problema e grafica la procedura

> **secanti\_graph(3,2,f,epsi,max\_iter,1,3) ;**

$$f := x \rightarrow x^2 - 3$$

1.732050808



[ ]  
[ ] >